math / data /lean /ComparatorChallenges /BiholderTransport.lean
introvoyz045's picture
Migrated from GitHub
4a56cd6 verified
Raw History Blame Contribute Delete
5.41 kB
import Mathlib
namespace OAI
noncomputable section
open Set MeasureTheory Manifold Bundle
open scoped ContDiff Manifold ENNReal NNReal Topology
namespace WeakMTWTransport
abbrev Model (n : β„•) := EuclideanSpace ℝ (Fin n)
section MetricDefinitions
variable {M : Type*} [MetricSpace M]
/-- Squared-distance cost with factor one half. -/
def cost (x y : M) : ℝ := dist x y ^ 2 / 2
/-- Calibration of Hausdorff measure to Euclidean volume. -/
def euclideanVolumeFactor (n : β„•) : ℝβ‰₯0∞ :=
volume (Metric.ball (0 : Model n) 1) /
(Measure.hausdorffMeasure (n : ℝ) : Measure (Model n))
(Metric.ball (0 : Model n) 1)
variable [MeasurableSpace M]
/-- Riemannian volume calibrated to Lebesgue measure in Euclidean space. -/
def metricVolume [BorelSpace M] (n : β„•) : Measure M :=
euclideanVolumeFactor n β€’ Measure.hausdorffMeasure (n : ℝ)
/-- Almost-everywhere measurable probability densities between fixed bounds. -/
def AdmissibleDensity (vol : Measure M) (lam cap : ℝ) (rho : M β†’ ℝ) : Prop :=
AEMeasurable rho vol ∧ Integrable rho vol ∧
(∫ x, rho x βˆ‚vol) = 1 ∧ (βˆ€α΅ x βˆ‚vol, lam ≀ rho x ∧ rho x ≀ cap)
/-- Multiplication by a density, ignoring exceptional negative values on null sets. -/
def densityMeasure (vol : Measure M) (rho : M β†’ ℝ) : Measure M :=
vol.withDensity (fun x => ENNReal.ofReal (rho x))
/-- Optimality for the Monge minimization problem. -/
def IsOptimalMap (vol : Measure M) (rho0 rho1 : M β†’ ℝ) (T : M β†’ M) : Prop :=
Measurable T ∧ Measure.map T (densityMeasure vol rho0) = densityMeasure vol rho1 ∧
βˆ€ S : M β†’ M, Measurable S β†’
Measure.map S (densityMeasure vol rho0) = densityMeasure vol rho1 β†’
(∫ x, cost x (T x) * rho0 x βˆ‚vol) ≀
(∫ x, cost x (S x) * rho0 x βˆ‚vol)
/-- The two global estimates have the very same exponent and constant. -/
def BiHolderEstimate (alpha C : ℝ) (T : M β‰ƒβ‚œ M) : Prop :=
(βˆ€ x x', dist (T x) (T x') ≀ C * (dist x x') ^ alpha) ∧
(βˆ€ y y', dist (T.symm y) (T.symm y') ≀ C * (dist y y') ^ alpha)
end MetricDefinitions
section RiemannianDefinitions
variable {n : β„•} {M : Type*} [MetricSpace M] [ChartedSpace (Model n) M]
[IsManifold π“˜(ℝ, Model n) ∞ M]
[RiemannianBundle (fun x : M => TangentSpace π“˜(ℝ, Model n) x)]
/-- Complete affinely parametrized geodesic with prescribed initial tangent.
Local metric minimization is an intrinsic definition of Riemannian geodesics. -/
def IsGeodesicWithInitialData (x : M) (v : TangentSpace π“˜(ℝ, Model n) x)
(gamma : ℝ β†’ M) : Prop :=
ContMDiff π“˜(ℝ, ℝ) π“˜(ℝ, Model n) ∞ gamma ∧
(⟨gamma 0, mfderiv π“˜(ℝ, ℝ) π“˜(ℝ, Model n) gamma 0 (1 : ℝ)⟩ :
TangentBundle π“˜(ℝ, Model n) M) = ⟨x, v⟩ ∧
βˆ€ t : ℝ, βˆƒ eps : ℝ, 0 < eps ∧
βˆ€ s u : ℝ, |s - t| < eps β†’ |u - t| < eps β†’
dist (gamma s) (gamma u) = |s - u| * β€–vβ€–
/-- Evaluation at time one of the geodesic determined by initial data. -/
def riemannianExp (x : M) (v : TangentSpace π“˜(ℝ, Model n) x) : M := by
classical
exact if h : βˆƒ gamma, IsGeodesicWithInitialData x v gamma then (Classical.choose h) 1 else x
/-- The open injectivity domain, defined by extendible minimizing vectors. -/
def injectivityDomain (x : M) : Set (TangentSpace π“˜(ℝ, Model n) x) :=
{p | βˆƒ a : ℝ, 1 < a ∧ dist x (riemannianExp x (a β€’ p)) = a * β€–pβ€–}
/-- The Ma–Trudinger–Wang fourth derivative with factor -3/2. -/
def mtw (x : M) (p xi eta : TangentSpace π“˜(ℝ, Model n) x) : ℝ :=
-(3 / 2 : ℝ) *
iteratedDeriv 2 (fun s : ℝ =>
iteratedDeriv 2 (fun t : ℝ =>
cost (riemannianExp x (t β€’ xi)) (riemannianExp x (p + s β€’ eta))) 0) 0
/-- Weak MTW only at interior velocities and only for orthogonal test vectors. -/
def WeakMTW : Prop :=
βˆ€ (x : M) (p : TangentSpace π“˜(ℝ, Model n) x), p ∈ injectivityDomain x β†’
βˆ€ xi eta : TangentSpace π“˜(ℝ, Model n) x,
inner ℝ xi eta = 0 β†’ 0 ≀ mtw x p xi eta
end RiemannianDefinitions
/-- Uniform bi-HΓΆlder optimal transport for all densities between fixed bounds
on a compact connected smooth Riemannian manifold satisfying weak MTW. -/
theorem uniform_biHolder_transport
{n : β„•} (hn : 2 ≀ n) {M : Type*}
[MetricSpace M] [CompactSpace M] [ConnectedSpace M]
[ChartedSpace (Model n) M] [IsManifold π“˜(ℝ, Model n) ∞ M]
[RiemannianBundle (fun x : M => TangentSpace π“˜(ℝ, Model n) x)]
[IsContMDiffRiemannianBundle π“˜(ℝ, Model n) ∞ (Model n)
(fun x : M => TangentSpace π“˜(ℝ, Model n) x)]
[IsRiemannianManifold π“˜(ℝ, Model n) M]
[MeasurableSpace M] [BorelSpace M]
(hmtw : WeakMTW (n := n) (M := M))
(lam cap : ℝ) (hlam : 0 < lam) (hcap : lam ≀ cap)
(hclass : βˆƒ rho : M β†’ ℝ, AdmissibleDensity (metricVolume n) lam cap rho) :
βˆƒ alpha C : ℝ, 0 < alpha ∧ alpha ≀ 1 ∧ 0 ≀ C ∧
βˆ€ rho0 rho1 : M β†’ ℝ,
AdmissibleDensity (metricVolume n) lam cap rho0 β†’
AdmissibleDensity (metricVolume n) lam cap rho1 β†’
βˆƒ T : M β‰ƒβ‚œ M,
IsOptimalMap (metricVolume n) rho0 rho1 T ∧
(βˆ€ S : M β†’ M, IsOptimalMap (metricVolume n) rho0 rho1 S β†’
S =ᡐ[densityMeasure (metricVolume n) rho0] T) ∧
BiHolderEstimate alpha C T := by
sorry
end WeakMTWTransport
end
end OAI